

# Building a Library of Mechanized Mathematical Proofs

Why do it?

And what is it like to do?

Rob Arthan


Lemma 1 Ltd. / Queen Mary, University of London

ICMS 2010

14th September, 2010

## A Hypothetical Question

What would Newton or the Bernoullis or Gauss or Delaunay or Hilbert or ...your list goes here ... have done with one of these:



Formalised Mathematics ca. 1910:  $1 + 1 = \text{hmmm?}$

*“The properties of  $\dot{2}$  are largely analogous to those of 1, while the properties of  $2_r$  are more analogous to those of 2.”*

A.N. Whitehead and B. Russell *Principia Mathematica* \*56 **p. 375** 1910

What about properties of  $e$  or  $\pi$  or  $\zeta(1)$  or  $\pi_{13}(S^5)$  or ...?

- The immense symbolic processing tasks defy human capabilities.
- But 100 years on we have machines to do all that.
- Machines don't know what symbols to process.
- Synergy between human and computer is what's called for.

## *Pure Mathematician: Why bother?*


- Mathematical arguments do contain errors!
- How many in the classification of finite simple groups?
- Even Aigner & Ziegler's "*Proofs from THE BOOK*" (ed. 1).

See Bill Casselman. *The Difficulties of Kissing in Three Dimensions*.

- Avoid controversies about use of computers:
  - 4 colour theorem
  - Kepler sphere packing conjecture
- Eliminate circle-squarers, cube-doublers, angle-trisectors!

...or even serious, but misguided "provers" of  $P = NP$

*Engineer: Why bother?*



A simple example: calculating bearings

- Specification: take  $\beta_{(x,y)}$  chosen in the appropriate quadrant where:

$$\tan \beta_{(x,y)} = \frac{y}{x}$$

- Design: calculate the bearing using:  $\beta = \tan^{-1}(y/x)$
- Code: `beta = atan(y/x)`
- BUT:  $\beta_{(1,1)} = \pi/4 \neq 5\pi/4 = \beta_{(-1,-1)}$

Testing tends not to find subtle errors.

## *Computer scientist: Why bother?*

Computer scientists...

- ...prove theorems
- ...specify, design and code programs

See above!

## Proposal

We should use computers to support mathematical endeavours.

- I will describe one approach that offers:
  - high assurance ( $1 \neq 2$ )
  - extensibility and programmability
  - access to a large library of existing results
- In the rest of the talk I hope to:
  - Show how the LCF paradigm offers high assurance
  - Give a flavour of what it is like to develop a theory like the calculus in an LCF-style system

## A Deductive System

Judgments:  $m \text{ D } n$ ,  $m, n \in \mathbb{Z}$  (intended meaning:  $m$  divides  $n$ )

Rules:

$$\frac{}{m \text{ D } m} : \text{axiom}, m \neq 0$$
$$\frac{m \text{ D } n}{m \text{ D } -n} : - \quad \frac{m \text{ D } n_1 \quad m \text{ D } n_2}{m \text{ D } n_1 + n_2} : +$$

A deduction:

$$\frac{\frac{}{1 \text{ D } 1} : \text{axiom} \quad \frac{1 \text{ D } 1}{1 \text{ D } -1} : -}{1 \text{ D } 0} : +$$

- Robin Milner's LCF method implements a deductive system as a data type.
- Strongly typed programming language enforces the rules.



## The Deductive System in ML

Demo 1

```
local
  datatype THEOREM = D of (int * int);
in
  type THEOREM = THEOREM;
  infix D; infix ++;
  exception NOT_ALLOWED;

  fun axiom m = if m <> 0 then m D m else raise NOT_ALLOWED;
  fun -- (m D n) = m D ~n;
  fun (m1 D n1) ++ (m2 D n2) =
    if m1 = m2 then m1 D (n1 + n2) else raise NOT_ALLOWED;
end;
```

## A Decision Procedure

Demo 1 concluded.

Given  $m$  and  $n$ , the function `decide` tries to prove that  $m$  divides  $n$ :

```
fun decide m n =  
  if n < 0 then --(decide m (~n))  
  else if n <= m then axiom m  
  else decide m (n-m) ++ axiom m;
```

Logical kernel will not allow invalid deductions:

```
> decide 2 6;  
val it = 2 D 6 : THEOREM  
> decide 2 7;  
val it = 2 D 8 : THEOREM  
> decide 0 0;  
Exception- NOT_ALLOWED raised
```

## A Real System: ProofPower-HOL

Demo 2.

- Member of the HOL family implemented for industrial use.  
Cf. Classic HOL (Gordon), HOL IV (Slind, Norrish), HOL Light (Harrison).
- Expressions and predicates represented by the type *TERM*,  
entered using “Quine corners”:  $\ulcorner 1 + 2 \urcorner$ ,  $\ulcorner 1 = 2 \urcorner$
- Abstract data type of theorems is the type *THM*. Printed with  
a turnstile:  $\vdash \neg 1 = 2$ .
- Extensive facilities for programming with syntax.
- Maintains a database of theories containing specifications (i.e.,  
defining properties of types and constants) and theorems.
- Powerful higher-level tools for automated and interactive proof.

## Characteristics of the LCF Approach

- Logical kernel is small and simple to check  
— possibly even formally verifiable.
- Derived rules may fail to produce desired output but can't produce unsound results.
- Can safely code very complex algorithms.
- Potential for cross-checking using alternative compilers and/or implementations.
- There are other complementary approaches.

## Definitions In ProofPower-HOL

- Syntax for defining new constants (including functions, functionals etc.) with an arbitrary defining property:

$$\frac{\langle NAME \rangle : \langle TYPE \rangle}{\langle DEFINING PROPERTY \rangle}$$

- Proof obligation to verify consistency (often discharged automatically).

HOL Constant

$$\mathbf{Even} : \mathbb{N} \text{ SET}$$

$$\forall n \bullet n \in \text{Even} \Leftrightarrow (\exists m \bullet n = 2 * m)$$

HOL Constant

$$\mathbf{Odd} : \mathbb{N} \text{ SET}$$

$$\forall n \bullet \neg 2 * n \in \text{Odd} \wedge 2 * n + 1 \in \text{Odd}$$

## Calculus In ProofPower-HOL 1

- Write  $(f \text{ Deriv } c) x$  to mean function  $f$  has derivative  $c$  at  $x$  (i.e.,  $df/dx = c$  or  $f'(x) = c$  in the usual vernacular).

$\$Deriv : (\mathbb{R} \rightarrow \mathbb{R}) \rightarrow \mathbb{R} \rightarrow \mathbb{R} \rightarrow \text{BOOL}$

---

$\forall f \ c \ x \bullet (f \text{ Deriv } c) x$

$\Leftrightarrow \forall e \bullet 0. < e \Rightarrow \exists d \bullet 0. < d \wedge$

$\forall y \bullet \text{Abs}(y-x) < d \wedge \neg y=x \Rightarrow \text{Abs}((f \ y - f \ x)/(y-x) - c) < e$

- This definition is trivially consistent.
- All the usual theorems: product rule, chain rule, Rolle, IVT, MVT, ....

## Calculus In ProofPower-HOL 2

- Define the exponential function by the differential equation:

$$\mathbf{Exp} : \mathbb{R} \rightarrow \mathbb{R}$$

$$\mathbf{Exp} 0. = 1. \wedge (\forall x \bullet (\mathbf{Exp} \mathbf{Deriv} \mathbf{Exp} x) x)$$

- Prove the consistency via theory of power series and differentiation of limits.
- Logarithm defined as left inverse of exponential.
- All the usual basic theorems, e.g.,

$$\vdash \forall x \bullet 0. < x \Rightarrow (\mathbf{Log} \mathbf{Deriv} x^{-1}) x : \mathbf{THM}$$

- Similar treatment of trigonometric functions.

## Calculus In ProofPower-HOL 3

Demo 2 continued.

- Integration via the Kurzweil-Henstock gauge integral. FTC, areas, ...
- A theorem from antiquity:

$$\vdash \forall r \bullet 0. < r \Rightarrow$$

$$\{(x, y) \mid \text{Sqrt } (x \wedge 2 + y \wedge 2) \leq r\} \text{ Area } \pi * r \wedge 2$$

- A theorem of Minkowski:

$$\vdash \forall A \bullet$$

$$A \in \text{Convex} \wedge A \in \text{Bounded} \wedge \neg A = \{\}$$

$$\wedge (\forall x \ y \bullet (x, y) \in A \Rightarrow (\sim x, \sim y) \in A)$$

$$\wedge A \text{ Area } a \wedge a > 4.$$

$$\Rightarrow \exists i \ j : \mathbb{Z} \bullet (\mathbb{Z}R \ i, \mathbb{Z}R \ j) \in A \wedge \neg(\mathbb{Z}R \ i, \mathbb{Z}R \ j) = (0., 0.)$$

- de Bruijn factor. Formal / Informal. Typically 0.5 – 5?
- See Freek Wiedijk's web site <http://www.cs.ru.nl/~freek/>



## Example: A Combinatorics Problem

Demo 2 concluded.

- $(m + n)^m$  should give the number of ways of drawing  $m$  samples with replacement out of a set of  $m + n$  elements.
- *DistinctSamples*  $n m$  should give the number of ways of drawing  $m$  samples without replacement out of a set of  $m + n$  elements.

*DistinctSamples* :  $\mathbb{N} \rightarrow \mathbb{N} \rightarrow \mathbb{N}$

$(\forall n \bullet \text{DistinctSamples } n \ 0 = 1)$

$\wedge (\forall m \ n \bullet$

$\text{DistinctSamples } n \ (m+1) = (n+m+1) * \text{DistinctSamples } n \ m)$

- Plan: give high assurance solution to a combinatorics problem by:
  - 1) Proving that these functions do give the desired results.
  - 2) Symbolically executing them inside the theorem prover.

## Final Remarks

- Machine-checked proof has a part to play.
- Technology can help: critical bugs are not inevitable!
- Formalisation need not lead to combinatorial explosion.
- Lots of fascinating work to do.

Thank you!

## Links

- For ProofPower

<http://www.lemma-one.com/ProofPower/index/index.html>

- Other implementations of HOL:

<http://hol.sourceforge.net/>

<http://www.cl.cam.ac.uk/research/hvg/Isabelle/>

<http://www.cl.cam.ac.uk/~jrh13/hol-light/>

- The Flyspeck project:

<http://code.google.com/p/flyspeck/>

- Freek Wiedijk's progress report on 100 theorems:

<http://www.cs.ru.nl/~freek/100/index.html>

- The Mizar project:

<http://mizar.uwb.edu.pl/>

The following is the Standard ML source of the demos:

Text dumped to file 82dedsys.ML

```
local
  datatype THEOREM = D of (int * int);
in
  type THEOREM = THEOREM;
  infix D; infix ++;
  exception NOT_ALLOWED;

  fun axiom m = if m <> 0 then m D m else raise NOT_ALLOWED;

  fun -- (m D n) = m D ~n;

  fun (m1 D n1) ++ (m2 D n2) =
    if m1 = m2 then m1 D (n1 + n2) else raise NOT_ALLOWED;
end;
```

Text dumped to file 82s1.ML

```
1+2;
fun fact n = if n <= 0 then 1 else n * fact(n-1);
fact 1s0;
map fact [1, 2, 3, 4, 5, 6, 7];
use "82dedsys.ML";
axiom 1;
1 D 1;
3 D 5;
val thm1 = axiom 1;
val thm2 = -- (axiom 1);
val thm3 = thm1 ++ thm2;
fun decide m n =
  if n < 0 then --(decide m (~n))
  else if n <= m then axiom m
  else decide m (n-m) ++ axiom m;
```

```

decide 1 1;
decide 1 ~1;
decide 2 6;
decide 13 1001;
decide 2 7;
decide 0 0;
decide ~1 1;

```

Text dumped to file 82s2.ML

```

val tm1 =  $\ulcorner 0 * 1 \urcorner$ ;
val ty1 = type_of tm1;
val (f1, args1) = strip_app tm1;
val ty_f1 = type_of f1;
val thm1 = get_spec f1;
val thm2 = list_∀_elim [  $\ulcorner 1 \urcorner$ ,  $\ulcorner 1 \urcorner$  ] thm1;
val thm3 = ∧_left_elim thm2;
val thm4 = rewrite_conv []  $\ulcorner (0+1+2+3)*(3+2+1+0) \urcorner$ ;
set_goal([],  $\ulcorner \forall m\ i\ j : \mathbb{N} \bullet m^{(i+j)} = m^i * m^j \urcorner$ );
a( REPEAT strip_tac );
a(induction_tac  $\ulcorner j \urcorner$ );
a(rewrite_tac[  $\mathbb{N}$ _exp_def ]);
a(asm_rewrite_tac[ plus_assoc_thm1 ,  $\mathbb{N}$ _exp_def ]);
a( PC_T1 "lin_arith" prove_tac [] );
val thm5 = pop_thm();
val tm2 =  $\ulcorner (\lambda x \bullet x + 1) \urcorner$ ;
val ty2 = type_of tm2;
val thm6 = rewrite_conv []  $\ulcorner (\lambda x \bullet x + 1) \ 2 \urcorner$ ;
val tm3 =  $\ulcorner [x; y; 1] \urcorner$ ;
val ty3 = type_of tm3;
val (f3, args3) = strip_app  $\ulcorner [x; y; z] \urcorner$ ;
val ty_f3 = type_of f3;
val thm7 = rewrite_conv[nth_def]  $\ulcorner Nth\ [x; y; z] \ 2 \urcorner$ ;
val thm8 = get_spec  $\ulcorner Map \urcorner$ ;
val ty4 = type_of  $\ulcorner Map \urcorner$ ;

```

```

val thm9 = rewrite_conv[map_def]  $\lceil \text{Map } (\lambda x \bullet x + 1) [1; 2; 3] \rceil$ ;
val tm4 =  $\lceil 2 \in \{1; 2; 3\} \rceil$ ;
val (f4, args4) = strip_app tm4;
val ty_f4 = type_of f4;
val thm4 = get_spec f4;
val thm10 = prove_rule[] tm4;
val thm11 = pc_rule1 "sets_ext1" prove_rule[elems_def]
   $\lceil \text{Elems } [1; 2; 3] = \{1; 2; 3\} \rceil$ ;
get_spec  $\lceil \pi \rceil$ ;
get_spec  $\lceil \text{Sin} \rceil$ ;
get_spec  $\lceil \$\text{Area} \rceil$ ;
get_spec  $\lceil \$\text{Int}_R \rceil$ ;
get_spec  $\lceil \text{Gauge} \rceil$ ;
get_spec  $\lceil \text{TaggedPartition} \rceil$ ;
get_spec  $\lceil \$\text{Fine} \rceil$ ;
get_spec  $\lceil \text{RiemannSum} \rceil$ ;
distinct_samples_def;
val thm12 = (print "\n\n"; conv_rule(MAP_C plus_conv)(pure_rewrite_conv
  [distinct_samples_def,
   pc_rule1 "lin_arith" prove_rule[]  $\lceil \forall m \bullet m * 1 = m \rceil$ ,
   plus_assoc_thm1]
   $\lceil \text{DistinctSamples } 10 (0+1+1+1) \rceil$ );
val thm13 = rewrite_rule[] thm12;
distinct_samples_finite_size_thm;
samples_finite_size_thm;
(* and then ... *)
use_file "82demo2proof.ML";
val thm14 = pop_thm();
get_spec  $\lceil \text{Elems} \rceil$ ;
get_spec  $\lceil \text{Distinct} \rceil$ ;
get_spec  $\lceil \text{Finite} \rceil$ ;

```

Text dumped to file 82demo2proof.ML

set\_goal([],  $\lceil (* \text{ Try not to show from HERE ... } *)$ )

```

let S = {L | Elems L ⊆ {i | 1 ≤ i ∧ i ≤ 365} ∧ # L = 23}
in let X = {L | L ∈ S ∧ ¬L ∈ Distinct}
in S ∈ Finite
  ∧ ¬#S = 0
  ∧ X ⊆ S
  ∧ #X / #S > 1/2
  ⌋);
a(rewrite_tac[let_def]);
a(strip_asm_tac(∀_elim⌈365⌋ range_finite_size_thm1));
a(lemma_tac⌈23 ≤ #{i | 1 ≤ i ∧ i ≤ 365}⌋ THEN1 asm_rewrite_tac[]);
a(all_fc_tac[distinct_samples_finite_size_thm]);
a(all_fc_tac[samples_finite_size_thm]);
a(REPEAT_N 2 (POP_ASM_T(ante_tac o ∀_elim⌈23⌋)) );
a(POP_ASM_T discard_tac THEN strip_tac THEN strip_tac);
a(pure_asm_rewrite_tac[conv_rule(ONCE_MAP_C eq_sym_conv)NR_one_one_thm,
  NR_N_exp_thm,
  R_frac_def]);
a(asm_tac(rewrite_conv[]⌈NR 365 ^ 23⌋));
a(PC_T1"predicates" rewrite_tac[] THEN strip_tac
  THEN1 (pure_asm_rewrite_tac[NR_one_one_thm]
    THEN PC_T1 "lin_arith" prove_tac[]));
a(REPEAT strip_tac THEN1 PC_T1 "sets_ext1" prove_tac[]);
a(pure_rewrite_tac[NR_one_one_thm]);
a(LEMMA_T⌈{L | (Elems L ⊆ {i | 1 ≤ i ∧ i ≤ 365} ∧ # L = 23)
  ∧ ¬L ∈ Distinct}
= {L | Elems L ⊆ {i | 1 ≤ i ∧ i ≤ 365} ∧ # L = 23} \
  {L | Elems L ⊆ {i | 1 ≤ i ∧ i ≤ 365} ∧ # L = 23
  ∧ L ∈ Distinct}⌋
  pure_rewrite_thm_tac
  THEN1 PC_T1 "sets_ext1" prove_tac[]);
a(lemma_tac⌈{L | Elems L ⊆ {i | 1 ≤ i ∧ i ≤ 365} ∧ # L = 23 ∧
  L ∈ Distinct} ⊆
  {L | Elems L ⊆ {i | 1 ≤ i ∧ i ≤ 365} ∧ # L = 23}⌋
  THEN1 PC_T1 "sets_ext1" prove_tac[]);
a(ALL_FC_T (MAP_EVERY (ante_tac o
  once_rewrite_rule[conv_rule(ONCE_MAP_C eq_sym_conv)NR_one_one_thm]))

```

```

[size_⊆_diff_thm]);
a(pure_rewrite_tac[NR_plus_homomorphism_thm,
  pc_rule1 "ℝ_lin_arith" prove_rule[]
  ⌈∀a b c:ℝ•a = b + c ⇔ b = a - c⌋
  THEN_STRIP_T pure_rewrite_thm_tac);
a(LIST_DROP_NTH_ASM_T [3, 6, 9] pure_rewrite_tac);
a(LEMMA_T ⌈∀a b c d:ℝ•NR 0 < b ∧ NR 0 < d ∧ a*d < b*c ⇒ a/b < c/d⌋
  bc_thm_tac
  THEN1 (REPEAT strip_tac
    THEN1 ALL_FC_T1 fc_⇔_canon asm_rewrite_tac[ℝ_cross_mult_less_thm]));
a(pure_asm_rewrite_tac[NR_N_exp_thm,
  pc_rule1 "ℝ_lin_arith" prove_rule[]
  ⌈∀a b c:ℝ•a < NR 2 * (b - c) ⇔ a + NR 2 * c < NR 2 * b⌋,
  REPEAT_C (once_rewrite_conv[distinct_samples_rw_thm]
    THEN_C rewrite_conv[])
  ⌈DistinctSamples (365 - 23) 23⌋]);
a(pure_rewrite_tac[NR_plus_homomorphism_thm1,
  NR_times_homomorphism_thm1,
  NR_less_thm]);
a(rewrite_tac[]) (* ... down to HERE! *);

```